Κεφάλαιο 7ο: Παραγώγιση και ολοκλήρωση.

7.1 Παραγώγιση και ολοκλήρωση

Όπως γνωρίζουμε η παράγωγος f'[x] *δίνεται από τον τύπο* Limit$\left[\frac{f(x+h)-f(x)}{h}, h \to 0\right]$ *παράδειγμα*:

```
Remove[f]
f[x_] := 3 x - 2 / x
Limit[(f[x + h] - f[x])/h, h → 0]
```

$3 + \frac{2}{x^2}$

Φυσικά υπάρχει και η συνάρτηση D[f[x],{x,n}] *που επιστρέφει χωρίς κόπο την* n-*ιοστή παράγωγο ως προς* x. Ειδικά *εάν* n=1(*πρώτη παράγωγος*) *τότε μπορούμε να γράψουμε* απλά D[f[x],x] ή *πιο* απλά *με* f'[x]. Με f''[x] *εννοούμε την* 2*η παράγωγο ως προς* x *κ.ο.κ.* Αν *θέλουμε να βρούμε την* n-*ιοστή παράγωγο χωρίς να μας ενδιαφέρει το όνομα της μεταβλητής που παραγωγίζουμε μπορούμε να χρησιμοποιήσουμε την*
Derivative[n][f]. Στο συγκεκριμμένο παράδειγμα της Derivative, *υποθέτουμε* ότι *η* f *έχει μόνο μια μεταβλητή*(*την* #1) *και ως προς αυτή και μόνο παραγωγίζουμε*. Το *αποτέλεσμα θα περιέχει αυτήν την ανώνυμη μεταβλητή*.

```
D[f[x], {x, 1}]
D[f[x], x]
f'[x]
Derivative[1][f]
(*εύρεση τιμών*)
D[f[x], {x, 1}] /. x → 3
f'[3]

D[f[x], {x, 1}][3](* δεν δουλευει!*)
```

$3+\frac{2}{x^2}$

$3+\frac{2}{x^2}$

$3+\frac{2}{x^2}$

$3+\frac{2}{\#1^2}$ &

$\frac{29}{9}$

$\frac{29}{9}$

$\left(3+\frac{2}{x^2}\right)$[3]

Για την παραγώγιση συναρτήσεων με παραπάνω απο μια μεταβλητές η διαδικασία είναι παρόμοια.Για παράδειγμα με D[f[x,y],{x,m},{y,n}] ***επιστρέφεται η μερική παράγωγος*** $\partial^{n+m} f / \partial x^m \partial y^n$ ***όπου πρώτα παραγωγίζουμε ως προς*** y(n-***ιοστή παράγωγος***) ***και μετά ως προς*** x(m-***ιοστή παράγωγος της προκύπτουσας συνάρτησης***). Η D[f,x,y] ***είναι συντομογραφία της*** D[f[x,y],{x,1},{y,1}] ***δηλ. της*** $\partial_{x,y} f$. Επίσης μπορούμε να χρησιμοποιήσουμε την Derivative[m,n][f] που είναι η αντίστοιχη της D[f[x,y],{x,m},{y,n}] όπου χρησιμοποιούμε ανώνυμες μεταβλητές. Η χρήση ανώνυμων μεταβλητών έχει το πλεονέκτημα ότι μπορούμε να τις αντικαταστήσουμε **άμμεσα(**χωρίς χρήση του ->) με τιμές ή μεταβλητές . Αυτό συμβαίνει διότι η Derivative έχει σαν αποτέλεσμα **μια συνάρτηση** (Function). Δείτε προσεκτικά και τα παραδείγματα παρακάτω

```
Remove[f]
f[x_, y_] := x^2 Sin[1/y]
D[f[x, y], {x, 1}, {y, 2}]
D[f[x, y], {x, 1}, {y, 2}] // FullForm
Derivative[1, 2][f]
Derivative[1, 2][f] // FullForm
D[f[x, y], {x, 1}, {y, 2}] /. {x → 2, y → 3}
D[f[x, y], {x, 1}, {y, 2}] /. x → 2 /. y → 3
 (*τιμή της D στο  σημείο (2,3) *)
D[f[x, y], {x, 1}, {y, 2}] /. {x, y} -> {2, 3}(*δεν δουλεύει σωστά*)
D[f[x, y], {x, 1}, {y, 2}][2, 3]
(*δεν δουλεύει σωστά διότι  η D δεν έχει έξοδο μια συνάρτηση!*)
Derivative[1, 2][f][2, 3]
(*τιμή της Derivative στο  σημείο (2,3) *)
```

$$2\,x\left(\frac{2\,\mathrm{Cos}[\frac{1}{y}]}{y^3}-\frac{\mathrm{Sin}[\frac{1}{y}]}{y^4}\right)$$

```
Times[2, x, Plus[Times[2, Power[y, -3], Cos[Power[y, -1]]],
  Times[-1, Power[y, -4], Sin[Power[y, -1]]]]]
```

$$2\,\#1\left(-\frac{\mathrm{Sin}[\frac{1}{\#2}]}{\#2^4}+\frac{2\,\mathrm{Cos}[\frac{1}{\#2}]}{\#2^3}\right)\ \&$$

```
Function[Times[2, Slot[1],
  Plus[Times[-1, Sin[Power[Slot[2], -1]], Power[Slot[2], -4]],
   Times[2, Cos[Power[Slot[2], -1]], Power[Slot[2], -3]]]]]
```

$$4\left(\frac{2}{27}\,\mathrm{Cos}\left[\frac{1}{3}\right]-\frac{1}{81}\,\mathrm{Sin}\left[\frac{1}{3}\right]\right)$$

$$4\left(\frac{2}{27}\,\mathrm{Cos}\left[\frac{1}{3}\right]-\frac{1}{81}\,\mathrm{Sin}\left[\frac{1}{3}\right]\right)$$

$$2\,x\left(\frac{2\,\mathrm{Cos}[\frac{1}{y}]}{y^3}-\frac{\mathrm{Sin}[\frac{1}{y}]}{y^4}\right)$$

$$\left(2\,x\left(\frac{2\,\mathrm{Cos}[\frac{1}{y}]}{y^3}-\frac{\mathrm{Sin}[\frac{1}{y}]}{y^4}\right)\right)[2,\ 3]$$

$$4\left(\frac{2}{27}\,\text{Cos}\left[\frac{1}{3}\right]-\frac{1}{81}\,\text{Sin}\left[\frac{1}{3}\right]\right)$$

Σχόλια: 1)Μπορούμε να παραγωγίσουμε και στην περίπτωση που η συνάρτηση μας προκύπτει από παρεμβολή κάποιων δεδομένων(παρεμβολή κάνουμε π.χ με την NDSolve). Στο επόμενο παράδειγμα η συνάρτηση rec προκύπτει απο τα τιμές του πίνακα t1(που περιέχει 10 τιμές της $\frac{1}{x}$) . Στην συνέχεια ορίζουμε την παράγωγο της με το γράμμα f (προσέξτε την χρήση του = αντί του := στον ορισμό. Το λόγο θα το δούμε στο επόμενο κεφάλαιο) και την σχεδιάζουμε. Για οπτικό έλεγχο για το πόσο καλά συμπεριφέρεται η rec σε σχέση με την $\frac{1}{x}$, κάνουμε και την γραφική παράσταση της παραγώγου της $\frac{1}{x}$.

```
t1 = Table[{x, 1/x}, {x, 1, 10}]
```

$$\left\{\{1,1\},\left\{2,\frac{1}{2}\right\},\left\{3,\frac{1}{3}\right\},\left\{4,\frac{1}{4}\right\},\left\{5,\frac{1}{5}\right\},\left\{6,\frac{1}{6}\right\},\left\{7,\frac{1}{7}\right\},\left\{8,\frac{1}{8}\right\},\left\{9,\frac{1}{9}\right\},\left\{10,\frac{1}{10}\right\}\right\}$$

```
rec = Interpolation[t1]
```

```
InterpolatingFunction[{{1, 10}}, <>]
```

```
f[x_] = D[rec[x], x]
```

```
Plot[f[x], {x, 1, 10}]
```




```
- Graphics -
```

```
Plot[Evaluate[D[1 / x, x]], {x, 1, 10}]
```




```
- Graphics -
```

Η Evaluate είναι πολύ χρήσιμη εδώ γιατί αναγκάζει το *Mathematica* πρώτα να υπολογίσει την D δηλ. την παράγωγο της 1/x πριν αρχίσει να κάνει οτιδήποτε άλλο (εδώ να σχεδιάσει μια γραφική παράσταση). Αν δεν μπεί το Evaluate τότε η Plot πάει και αντικαθιστά τιμές του x στην παράγωγο χωρίς να την έχει υπολογίσει με αποτέλεσμα να μην μπορεί να γίνει η γραφική παράσταση.

2)Ας δούμε και ένα παράδειγμα παραγώγισης μιας συνάρτησης που ορίζεται **με περιπτώσεις**

```
Remove[g]
g[x_, y_] := If[y ≥ 0, y * x^2, x Sin[1 / y]]
D[g[x, y], {y, 1}]
```

$$\text{If}\left[y \geq 0,\ x^2,\ -\frac{x\,\text{Cos}\left[\frac{1}{y}\right]}{y^2}\right]$$

Προσοχή! το αποτέλεσμα δεν είναι σωστό διότι δεν υπάρχει η D[g[x,y],y] στο σημείο (1,0) ! όπως εύκολα το ελέγχουμε...σχεδιάζοντας την $\frac{g[1,h]-g[1,0]}{h} = \frac{\text{Sin}[\frac{1}{h}]}{h}$ για h αρνητικό!

```
Plot[Sin[1 / h] / h, {h, -1, 0}]
```




```
- Graphics -
```

Θα πρέπει λοιπόν να είμαστε ιδιαίτερα προσεκτικοί.. Πάντως για όλες τις άλλες τιμές (x,y) όπου το y είναι διαφορετικό του 0 τα πράγματα είναι απλά, και οι μερικές παράγωγοι υπολογίζονται εύκολα π.χ για y <0 έχουμε...

```
Remove[g]
g[x_, y_] := x Sin[1 / y]
D[g[x, y], {x, 1}]
D[g[x, y], {y, 1}]
D[g[x, y], {x, 1}, {y, 1}]
D[g[x, y], {y, 1}, {x, 1}]
D[g[x, y], y, x]
D[g[x, y], y, x][3, 2](*δεν δουλεύει*)
D[g[x, y], y, x] /. x → 3 /. y → 2
```

$$\text{Sin}\left[\frac{1}{y}\right]$$

$$-\frac{x\,\text{Cos}\left[\frac{1}{y}\right]}{y^2}$$

$$-\frac{\text{Cos}\left[\frac{1}{y}\right]}{y^2}$$

$$-\frac{\text{Cos}\left[\frac{1}{y}\right]}{y^2}$$

$$-\frac{\text{Cos}\left[\frac{1}{y}\right]}{y^2}$$

$$\left(-\frac{\text{Cos}\left[\frac{1}{y}\right]}{y^2}\right)[3,\,2]$$

$$-\frac{1}{4}\,\text{Cos}\left[\frac{1}{2}\right]$$

3) Όπως είπαμε και προηγουμένως θα χρειαστούμε να χρησιμοποιήσουμε το σύμβολο της αντικατάστασης /. για να θέσουμε τιμές στα x,y για την παράγωγο. Αν θέλουμε **να αποφύγουμε** το σύμβολο /. θα μπορούσαμε να χρησιμοποιήσουμε την Derivative όπως είδαμε ή να ορίσουμε την ίδια την παράγωγο με ένα νέο σύμβολο συνάρτησης (με χρήση του = και όχι του := Το = είναι διαφορετικό απο το := όπως θα δούμε και σε επόμενα κεφάλαια). Δείτε τα παραδείγματα που ακολουθούν:

```
merikhparagwgos[x_, y_] = D[g[x, y], y, x]
merikhparagwgos[3, 2]
```

$-\frac{\text{Cos}[\frac{1}{y}]}{y^2}$

$-\frac{1}{4}\text{Cos}[\frac{1}{2}]$

4) H D[f[x],{x,4}] *παράγει το ίδιο αποτέλεσμα με το* f''''[x], *με το* D[f[x],x,x,x,x] *και τελος με το* $\partial_{\{x,4\}} f[x]$. *Διαλέξτε* ότι *σας αρέσει*! Το *σύμβολο* $\partial_{\square}$ □ *το βρίσκουμε στην παλέττα* BasicInput. Εδώ *πρέπει να προσέχουμε* όταν *έχουμε γινόμενο συναρτήσεων να βάζουμε παρενθέσεις αλλιώς υπάρχει περίπτωση λάθους απο δική μας υπαιτιότητα π.χ αν ξεχάσουμε τις παρενθέσεις στην* ∂_x (x Sin[x])*θα πάρουμε κατα λάθος*

```
∂x x Sin[x]
```

```
Sin[x]
```

δηλ.παραγώγισε μόνο την x *και* όχι *το γινόμενο* x Sin[x]!

5)Άν *θελουμε να παραγωγίσουμε την* g[y] *ως προς* x *τότε* όπως *γνωρίζουμε απο την* Θεωρία *η παράγωγος* είναι 0 *εκτός και αν θεωρσύμε* ότι *η* y *δεν* είναι *μια σταθερά* **αλλά** *μια συνάρτηση του* x. Αυτό *δηλώνεται με την συνάρτηση* NonConstants *π.χ*

```
Remove[g]
D[g[y], x, NonConstants → y]
Remove[g, f]
D[f[g[t]], x, NonConstants → t](*κανόνας αλυσίδας*)
```

```
D[y, x, NonConstants → {y}] g'[y]
```

```
D[t, x, NonConstants → {t}] f'[g[t]] g'[t]
```

Στο δεύτερο παράδειγμα θεωρήσαμε ότι *η μεταβλητή* t *ειναι μια συνάρτηση της* x.

7.2 Εύρεση τοπικών ακρότατων με χρήση των μερικών παραγώγων

Έχουμε ήδη αναφέρει την συνάρτηση FindMinimum *για την εύρεση κάποιων ακροτάτων μιας συνάρτησης* f. *Όμως εδώ θα προσπαθήσουμε να βρούμε τα τοπικά ακρότατα* ή *τουλάχιστον τα* "*πιθανά*" *τοπικά ακρότατα της* f *λύνοντας την εξίσωση* f'[x]==0 *αν η* f *έχει μόνο μια μεταβλητή* ή *ένα σύστημα εξισώσεων αν* έχει *παραπάνω* (*π.χ* $\partial_x f = \partial_y f = 0$ *αν η* f *έχει δύο μεταβλητές* την x,y). *Για την λύση των συστημάτων αυτών μπορούμε να χρησιμοποιήσουμε την* Solve. *Αν οι παράγωγοι δεν είναι πολυώνυμα τότε προτιμάμε να πάρουμε αριθμητικές λύσεις με την* NSolve ή *την* FindRoot. Εδώ *να θυμήσουμε ότι η* FindRoot *έχει την μορφή* `FindRoot[{`eqn_1, eqn_2, ... }, {x, x_0}, {y, y_0}, ...] *όπου* eqn_1, eqn_2, ... *είναι το σύστημα των εξισώσεων και* x_0, y_0, ... *είναι κάποιοι αριθμοί* "*κοντά*" *σε κάποια πραγματική λύση* (x,y,...) *του συστήματος* ή την μορφή `FindRoot[{`eqn_1, eqn_2, ... }, {x, x_0, xmin, xmax}, {y, y_0, ymin, ymax}, ...] όπου έχουμε θέσει και περιορισμούς στα x, y, ... δηλ. ζητάμε επιπλέον να ικανοποιούνται και οι xmin<x<xmax κ.ο.κ. ***Αυτό είναι και το μεγάλο πρόβλημα με την*** FindRoot: *πρέπει να ψάξεις κοντά αλλιώς μπορεί να μην βρεις καμμία λύση*! Παραδείγματα: *Ας προσπαθήσουμε να βρούμε τα τοπικά ακρότατα της* x Sin[1/y] *με την* NSolve *και την* FindRoot όπου έχουμε αντικαταστήσει την x με μια τιμή π.χ. x=0.05

```
Remove[g, g2]
g[x_, y_] := x * Sin[1 / y]
g2[x_, y_] := D[x * Sin[1 / y], y]
g2[x, y]
z = NSolve[g2[0.05, y] == 0, y]

g2[0.05, y] /. z(*Δοκιμή*)
0.05 Sin[1 / y] /. z
```

$$-\frac{x\,\mathrm{Cos}\left[\frac{1}{y}\right]}{y^2}$$

```
Solve::ifun :
 Inverse functions are being used by Solve, so some solutions may not be found.
```

{{y → -0.63662}, {y → 0.63662}}

$\{-7.55399\times10^{-18}, -7.55399\times10^{-18}\}$

{-0.05, 0.05}

```
ymin = -1; ymax = 1;
FindRoot[Evaluate[g2[0.05, y] == 0], {y, 0.1, ymin, ymax}]
```

```
{y → 0.0909457}
```

Να εξηγήσουμε ότι η NSolve *έβγαλε πολύ εύκολα ένα τοπικό ακρότατο στο σημείο* y=0.63662(*και άλλο ένα στο* y→-0.63662) . Είναι *μέγιστο διότι στο προηγούμενο κεφάλαιο είχαμε κάνει και την γραφική παράσταση της* g2[0.05,y]. *Δυστυχως δεν τα βρήκε* όλα. Με *την* **FindRoot** *μπορούμε να βρούμε και άλλα*! *Το προβλημα είναι πως διαλέγουμε το αρχικό μας* y0 *για να ξεκινήσει το ψάξιμο*. *Θα πρέπει να κάνουμε διάφορες δοκιμές στο* y0 *και για διάφορα διαστήματα* [ymin,ymax]. *Άλλο ένα μικρότερο πρόβλημα*(για τις νέες εκδόσεις του *Mathematica* δεν υπάρχει πρόβλημα): *πρέπει να γράψουμε τον τύπο της* g2 με Evaluate *αλλιώς βγαίνουν περίεργα μηνύματα όπως* το `FindRoot::frnum : Function {∂0.1(-0.0272011)}`is not a length 1 list of numbers at {y} = {0.1}. Το FindRoot μπερδεύεται με το D που εμφανίζεται μέσα στην g2. Το Evaluate αναγκάζει το g2 να υπολογιστεί πρώτο και άρα πρώτα υπολογίζεται η συνάρτηση το D[x*Sin[1/y],y] πριν γίνει οποιαδήποτε αντικατάσταση στην g2 κάποιας τιμής της y ή πριν γίνει οποιαδήποτε άλλη ενέργεια.. Για την Evaluate θα το συναντήσουμε και σε άλλες ενότητες. Αν θέλουμε να **αποφύγουμε** το Evaluate θα πρέπει να ορίσουμε το τύπο της g2 απευθείας (χωρίς την χρήση του Evaluate) ή να χρησιμοποιήσουμε την Derivative που μας δίνει έξοδο συνάρτηση

```
Remove[g2, g3]
g2[x_, y_] := -(x Cos[1/y])/y^2
g3 = Derivative[0, 1][g]
ymin = -1;
ymax = 1;
z = FindRoot[g2[0.05, y] == 0, {y, 0.1, ymin, ymax}]
FindRoot[g3[0.05, y] == 0, {y, 0.1, ymin, ymax}]

g[0.05, y] /. z
```

$$-\frac{\mathrm{Cos}\left[\frac{1}{\#2}\right]\,\#1}{\#2^2}\ \&$$

```
{y → 0.0909457}
```

```
{y → 0.0909457}
```

```
-0.05
```

Σχόλιο: Επαναλαμβάνουμε *ξανά την γενική αρχή*: *όταν ορίζουμε συναρτήσεις* αν δεν θελουμε να εμφανίζονται διάφορα μηνύματα λάθους για κάποιες εκδόσεις του *Mathematica*, *να μην χρησιμοποιούμε άλλες ήδη κατασκευασμένες αλλά να τις ορίζουμε "κατευθείαν"*. *Έτσι δεν θα ήταν καλό, για* *παράδειγμα, να ορίζαμε* g[x_,y_]:=x*Sin[1/y] *και στην συνέχεια* g2[x_,y_]:=D[g[x,y],y] *αλλά θα* ήταν *προτιμότερο να ορίζαμε κατ'ευθείαν* g2[x,y]:=$-\frac{x \operatorname{Cos}[\frac{1}{y}]}{y^2}$.

Αν θέλουμε περισσότερες λύσεις με μεγαλύτερη ακρίβεια θα πρέπει να αλλάξουμε την αρχική τιμή του y0 *να θέσουμε μικρότερα διαστήματα* ymin, ymax *και να βάλουμε επιπλέον ορίσματα μέσα στην* FindRoot *π.χ*

```
z = FindRoot[g2[0.05, y] == 0, {y, 0.01, 0.001, 0.09},
  AccuracyGoal → 24, WorkingPrecision → 34,
    MaxIterations → 50]
g[0.05, y] /. z

{y → 0.01010507575186637052500849291254059}
```

```
-0.05
```

Για πληροφορίες για τις AccuracyGoal κ.λ.π. ανατρέξτε στο Help.

Άσκηση: Χρησιμοποείστε την FindRoot με τις εξισώσεις D[g2[x,y],y==0 και D[g2[x,y],x]==0 (το = να *είναι διπλό)*για να βρείτε υποψήφια ακρότατα της g γύρω απο το σημείο (0.05, 0.05) ή όποιο άλλο εσείς θέλετε...

Άσκηση: *Δίνεται η συνάρτηση* f[x,y]:=x^2+ x*y+y^2-2*x-6*y. *Να βρείτε τα κριτικά σημεία της δηλ. τα σημεία που μηδενίζονται οι μερικές παράγωγοι. Στην συνέχεια χρησιμοποιώντας την* FindMinimum *προσπαθήσετε να απαντήσετε αν τα ακρότατα που βρήκατε είναι τοπικά μέγιστα* ή *τοπικά ελάχιστα.*Τέλος *χρησιμοποιείστε την* ContourPlot *για την* f *και για κατάλληλα διαστήματα* {xmin,xmax} *και* {ymin,ymax}*στους άξονες* Ox *και* Oy *αντίστοιχα (που να περιέχουν το* ή *τα σημεία που βρήκατε) έτσι ώστε να δείτε αν πράγματι η γραφική παράσταση που παίρνεται συμφωνεί με τα αποτελέσματα που βγάλατε.*(Προσοχή *πριν καλέσεται την* ContourPlot[x^2+x*y+y^2-2*x--6*y,{x,xmin,xmax},{y,ymin,ymax}] *να γράψετε* <<Graphics`Graphics3D` *διότι κάθε συνάρτηση για γραφικές παραστάσεις έχει το δικό της πακέτο εκτός από μερικές που δεν χρειάζεται να καλέσουμε το πακέτο τους*, διότι *φορτώνεται μόλις ανοίξουμε το Mathematica*). Με *το* Help *προσπαθήστε να μάθετε περισσότερα για την* ContourPlot *ώστε να κάνετε το γράφημα σας ελκυστικό.(π.χ δώστε χρώμα βάζοντας μέσα στην* ContourPlot *το* ColorFunction→Hue. *Στο ίδιο πακέτο ανήκει και η* ShadowPlot3D *επίσης με πολύ ωραία αποτελέσματα*). Η Plot3D(*αυτή δεν χρειάζεται κανένα ιδιαίτερο πακέτο*) *στην περίπτωση μας δεν βοηθά πολύ, εκτός και αν αλλάξετε λίγο τον χρωματισμό της επιφάνειας που προκύπτει με το επιπλέον χαρακτηριστικό* `ColorFunction → Hue` *μεσα στην* Plot3D.

7.3 Αόριστα ολοκληρώματα

Η *βασική εντολή για να βρούμε το αόριστο ολοκλήρωμα ως προς* x *είναι η* Integrate[f[x],x]. *π.χ*

```
Integrate[f'[x], x]
D[Integrate[f[x], x], x]

f[x]
```

```
f[x]
```

Δηλαδή το ολοκλήρωμα της παραγώγου της f *ως προς* x *είναι η* ίδια *η* f! Και *η παράγωγος του ολοκληρώματος της* f *ως προς* x είναι *πάλι η* f. Με άλλα *λόγια η ολοκλήρωση με την παραγώγιση* είναι *αντίστροφες συναρτήσεις.*Ας *κάνουμε και* άλλα *παραδείγματα*:

```
Remove[g]
Integrate[Log[x], x]
Integrate[Log[x], x] /. x → 2
g = (Integrate[Log[x], x] /. x → #) &
g[2]

Integrate[x^2 Cos[n x], x]
∫x^n dx
Integrate[Log[x y], x, y]
Integrate[Cos[Sin[x]], x]
∫E^(1-x^2) dx
```

```
-x + x Log[x]
```

```
-2 + 2 Log[2]
```

```
∫Log[x] dx /. x → #1 &
```

```
-2 + 2 Log[2]
```

$$\frac{2\,x\,\mathrm{Cos}[n\,x]}{n^2}+\frac{(-2+n^2\,x^2)\,\mathrm{Sin}[n\,x]}{n^3}$$

$$\frac{x^{1+n}}{1+n}$$

```
-2 x y + x y Log[x y]
```

```
∫Cos[Sin[x]] dx
```

$$\frac{1}{2}\,e\,\sqrt{\pi}\,\mathrm{Erf}[x]$$

Ας αρχίσουμε να κάνουμε παρατηρήσεις:

1)***Στο αόριστο ολοκλήρωμα*** το *Mathematica* ***παραλείπει*** ***τη σταθερά στο τέλος*** του ολοκληρώματος!

2)Με το /.x-> ... μπορούμε να βρούμε κάποια τιμή του αόριστου ολοκληρώματος στο οποίο η μεταβλητή που ολοκληρώνουμε είναι η **x**.

3)Με (Integrate[f[x],x]/.x→#)& μπορούμε να ορίσουμε το αόριστο ολοκλήρωμα $\int \mathbf{f}\, d\mathbf{x}$ **ως μια συνάρτηση** μιας μεταβλητής και να βρούμε εύκολα μια τιμή της. Πιο γενικά και η ίδια η προς ολοκλήρωση συνάρτηση f μπορεί να θεωρηθεί ως μεταβλητή(π.χ η #1) με την (Integrate[#1[x],x]/.x→#2)&. Με αυτό τον τρόπο μπορούμε να ολοκληρώνουμε εύκολα μια οποιαδήποτε συνάρτηση και να βρίσκουμε την τιμή της. π.χ (Integrate[#1[x],x]/.x→#2)&[Log,2].

4)*Μπορούμε να ολοκληρώσουμε ακόμα και αν η συνάρτηση μας έχει μια* ή *παραπάνω παραμέτρους* (*πχ την* n).

5) Τα *αόριστα ολοκληρώματα υπολογίζονται με την* ***σιωπηρή υπόθεση*** ότι *οι χρησιμοποιούμενες παράμετροι είναι τέτοιες* ώστε *όσο η ολοκληρωτέα συνάρτηση* όσο *και το αποτέλεσμα να* έχουν *έννοια*. Έτσι, *για παράδειγμα, το ολοκλήρωμα* $\frac{x^{1+n}}{1+n}$ *που επιστρέφει το* *Mathematica* *για την ολοκλήρωση* $\int x^{\wedge} n \, dx$ *ξέρουμε* ότι *είναι αληθές*, υπό *την προυπόθεση* ότι *το* n *είναι διαφορετικό του* -1, *αφού, ως γνωστό*, $\int x^{-1} dx$=ln[x].

6)Το *σύμβολο της ολοκλήρωσης* $\int \square \, d\square$ *μπορούμε να το βρούμε απο τις παλέτες* 3BasicInput. *Ένας* άλλος *τρόπος χωρίς παλέτες είναι να πατήσουμε διάφορους συνδυασμούς πλήκτρων π.χ για να γράψουμε το ορισμένο* $\int_0^7 f[x] \, dx$ *θα πρέπει να πατήσουμε*: ESCintESC CTRL[+] 0 CTRL[%] 7] CTRL[ ] f[x]ESCddESCx (*τα* [*και*] *δεν τα κτυπάμε*! *το* CTRL[+]*για παράδειγμα σημαίνει* ότι *πατάμε μαζί το* CTRL *και το* + *ενώ το* CTRL[ ] *είναι το* CTRL *μαζί με το* SPACE *πλήκτρο*) .

7)*Μπορούμε να* έχουμε *και διπλά* (*αλλά και* *πολλαπλά ολοκληρώματα*) όπως *για παράδειγμα το* Integrate[Log[x y],x,y] *που ισοδυναμει με* $\int dx \int dy \, \text{Log}[x, y]$*δηλ. πρώτα ολοκληρώνουμε ως προς* y *και το αποτέλεσμα μετά ως προς* x.

8) *Υπάρχει ένα μεγάλο πληθος ολοκληρωμάτων, τα οποία υπολογίζεται με την χρήση ειδικών συναρτήσεων* όπως *οι* Erf, EulerGamma, Fresnel, Hypergeometric, Elliptic *και* άλλες. *Μπορείται να βρείτε πληροφορίες στο* help.

9) *Υπάρχουν ολοκληρώματα τα οποία δεν μπορεί να υπολογίσει το* *Mathematica*. ***Σε μια*** *τέτοια περίπτωση επιστρέφεται το ίδιο το ολοκλήρωμα π.χ στο* $\int \text{Cos}[\text{Sin}[x]] \, dx$. ***Φυσικά*** *άν αντί αόριστο* έχουμε ***ορισμένο*** *ολοκλήρωμα τότε μπορούμε να πάρουμε μια αριθμητική* τιμή *του με την χρήση της* N ή *της* NIntegrate όπως *θα δούμε παρακάτω. π.χ*

```
olokl = NIntegrate[Cos[Sin[x]], {x, 0, Pi / 3}]

0.89975
```

10)Στην περίπτωση που η προς προς ολοκλήρωση συνάρτηση ορίζεται με περιπτώσεις (με χρήση της If) τότε σε μερικές περιπτώσεις η Integrate δεν μπορεί να δώσει απάντηση π.χ.(στο *Mathematica* v5. έχουμε σαφώς βελτίωση του προβλήματος που ανεφέρεται εδώ με την Integrate)

```
Remove[g]
g[x_, y_] := If[y ≥ 0, y*x^2, x Sin[1/y]]
Integrate[g[x, 3], x](* θέτουμε y=3 και ολοκληρώνουμε*)
Integrate[g[x, y], x](* εδώ δεν δίνει απάντηση*)
```

x^3

$\int \text{If}\left[y \geq 0,\ y\,x^2,\ x\,\text{Sin}\left[\frac{1}{y}\right]\right] dx$

11)Η Integrate μπορεί να χρησιμοποιηθεί ακόμα και στην περίπτωση που η συνάρτηση μας έχει βρεθεί με την μέθοδο της παρεμβολής(π.χ μετά απο χρήση της NDSolve). Στο παρακάτω παράδειγμα η rec είναι η συνάρτηση που είχαμε βρεί πιο πάνω. Η απάντηση (δηλ. το ολοκλήρωμα) έχει υπολογιστεί και αυτό με την μέθοδο της παρεμβολής...

```
f = Integrate[rec[x], x]
f /. x → 3
```

```
InterpolatingFunction[{{1, 10}}, <>][x]
```

$\frac{10}{9}$

7.4 Ορισμένα ολοκληρώματα

Η *βασική εντολή για τα ορισμένα ολοκληρώματα είναι όπως και στα αόριστα με την εισαγωγή των ορίων π.χ το ορισμένο ολοκλήρωμα* $\int_0^{Pi/3} \text{Sin}[x]^2\,\text{Cos}[x]^3\,dx$ *απλά θα το γράψουμε*

```
Integrate[Sin[x]^2 Cos[x]^3, {x, 0, Pi/3}]
```

$\frac{11\sqrt{3}}{160}$

Μπορούμε φυσικά να χρησιμοποιήσουμε την 3BasicInput *παλέτα ή πατώντας πλήκτρα.Μπορούμε φυσικά να βάλουμε και το άπειρο σε ένα ή και στα δύο άκρα*

```
Integrate[Sin[x]^2 Cos[x]^3, {x, 0, ∞}]
```

```
Integrate::idiv : Integral of Cos[x]^3 Sin[x]^2 does not converge on {0, ∞}.
```

$\int_0^{\infty} \text{Cos}[x]^3\,\text{Sin}[x]^2\,dx$

Σε αυτήν την περίπτωση το ολοκλήρωμα δεν υπάρχει! Σε *άλλη περίπτωση* όμως *υπάρχει π.χ*

```
Integrate[E^(-x^2), {x, 0, ∞}]
```

$$\frac{\sqrt{\pi}}{2}$$

Επίσης σε άλλες περιπτώσεις το Mathematica μπορεί να μας απαντήσει με το If *π.χ*

```
Integrate[x^n, {x, 0, 1}]
```

$$\text{If}\left[\text{Re}[n] > -1,\ \frac{1}{1+n},\ \int_0^1 x^n\, dx\right]$$

Αυτό σημαίνει ότι *εάν το πραγματικό* μέρος *της παραμέτρου* n (*διότι μπορεί* κάποιος *να* δώσει *και μιγαδική τιμή στο* n) είναι > *του* -1 *τότε το ολοκλήρωμα* είνα *ίσο με* $\frac{1}{1+n}$ *αλλιώς δεν μπορεί να δώσει απάντηση*! Π.*χ ας βάλουμε* n=2

```
% /. n → 2
```

$$\frac{1}{3}$$

Γενικά αν επιθυμούμε να θέσουμε κάποιους περιορισμούς στον υπολογισμό του ολοκληρώματος θα πρέπει να τις εισάγουμε με την εντολή θα Assumptions-> ... *οι περιορισμοί π.χ αν θελουμε να δηλώσουμε* ότι *μια παράμετρος* m *παίρνει πραγματικές τιμές* (*και όχι μιγαδικές*) *θα μπορούσαμε να το δηλώσουμε λέγοντας* ότι *το φανταστικό μέρος* είναι 0 : Im[m]==0 *π.χ*

```
Integrate[Sin[m x] / x, {x, 0, Infinity}, Assumptions → Im[m] == 0]
```

$$\frac{1}{2}\pi\,\text{Sign}[m]$$

Αν δεν βάζαμε τον περιορισμό αυτό θα παίρναμε:

```
Integrate[Sin[m x] / x, {x, 0, Infinity}]
```

$$\text{If}\left[\text{Im}[m] == 0,\ \frac{1}{2}\pi\,\text{Sign}[m],\ \int_0^\infty \frac{\text{Sin}[m\,x]}{x}\, dx\right]$$

Η *εντολή* NIntegrate[f[x],{x,a,b}] *επιστρέφει μια αριθμητική προσέγγιση του ορισμένου ολοκληρώματος* $\int_a^b f(x)\,dx$ *ενώ υπάρχουν και χρήσιμα επιπλέον χαρακτηριστικά που μπορούμε να προσθέσουμε* όπως *για παράδειγμα* AccuracyGoal-> 20 *για να βελτιώσουμε την ακρίβεια των υπολογισμων*. Π.*χ*

```
NIntegrate[Sin[x]^2 Cos[x]^3, {x, 0, Pi/3},
 AccuracyGoal → 20, WorkingPrecision → 30]

0.119078493020360313930
```

Η *συνάρτηση* NIntegrate *είναι χρήσιμη* όταν *η* Integrate *δεν μπορεί* να *μας βγάλει* ένα *αποτέλεσμα*. Η Integrate *κάνει συμβολικούς υπολογισμούς* ενώ *η* NIntegrate *χρησιμοποιεί* ***προσεγγιστικές*** *αριθμητικές μεθόδους*. Π.χ

```
(*Integrate[Abs[x-Log[x]+1.5],{x,1,3}]*)
NIntegrate[Abs[x - Log[x] + 1.5], {x, 1, 3}]

5.70416
```

Βάλαμε σε σχόλια το πρώτο ολοκλήρωμα διότι σε μας(στο δικό μας μηχάνημα) *το Mathematica* v4 *δεν μπόρεσε να βγάλει κάποιο αποτέλεσμα παρόλου που περιμέναμε αρκετή ώρα! Όμως πολύ γρήγορα η* NIntegrate *υπολόγισε το ολοκλήρωμα*! Επαναλαμβάνουμε ότι σε νεώτερες εκδόσεις του *Mathematica* η Integrate σαφώς συμπεριφέρεται καλύτερα.

7.5 Πολλαπλά ολοκληρώματα

Με *την εντολή* Integrate *μπορούμε να υπολογίσουμε και διπλά αλλά και τριπλά κ.ο.κ ολοκληρώματα αρκεί να διαμορφώσουμε κατάλληλα τα όρια* D *του ολοκληρώματος*. *Ας δούμε μερικές περιπτώσεις*

1)Η *περιοχή* D *αποτελείται απο κάποια διαστήματα π.χ* $D = \{(x, y) \in \mathbb{R}^2 / a \leq x \leq b,\ c \leq y \leq d\}$. *Τότε υπολογίζουμε τα ολοκληρώματα θέτοντας τα κατάλληλα όρια* {x,,a,b}, {y,c,d} *π.χ*

```
Integrate[x y^2 z^3 , {x, 0, 1}, {y, 0, 1}, {z, 0, 2}]

2/3
```

Σε *αυτήν την περίπτωση δεν παίζει ρόλο η **σειρά** ολοκλήρωσης*:

```
Integrate[x y^2 z^3 , {z, 0, 2}, {x, 0, 1}, {y, 0, 1}]

2/3
```

2) Η *περιοχή* D *είναι μια περιοχή του διδιάστατου χώρου που περικλείται απο κάποιες καμπύλες ή μια περιοχή του τρσδιάστατου χώρου που περικλείεται απο κάποιες επιφάνειες κ.ο.κ τότε θα πρέπει να προσέξουμε την **σειρά** που ολοκληρώνουμε(δηλ. ως προς ποιά μεταβλητή θα ολοκληρώσουμε πρώτα, μετά ποια ακολουθεί κ.ο.κ)*. *Ακολουθούν μερικά παραδείγματα* D *και συναρτήσεων* f *και η σειρά που ολοκληρώνουμε*. *Για να γίνουν και πιο παραστατικά θα κάνουμε **πρώτα** την γραφική παράσταση των*

χωρίων D. Για αυτό το σκοπό, ***στην διδιάστατη περίπτωση των*** D ***θα χρειαστεί να καλέσουμε το πακέτο*** Graphics`FilledPlot`

```
<< Graphics`FilledPlot`
FilledPlot[{x^2, -x^2}, {x, 0, 3}]
```




```
- Graphics -
```

εδώ $D = \{(x, y) \in \mathbb{R}^2 / 0 \le x \le 3,\ -x^2 \le y \le x^2\}$ ***και ολοκληρώνουμε μια*** f ***π.χ την*** 2 x -3 y^2 ***πρώτα ως προς*** y ***και μετά ως προς*** x ***δηλ.***

```
Integrate[2 x - 3 y^2, {x, 0, 3}, {y, -x^2, x^2}]
```

$$-\frac{3807}{7}$$

Πρέπει να προσέξουμε ότι το ολοκλήρωμα που πρέπει να υπολογιστεί πρώτο, μπαίνει πάντοτε στο τέλος του Integrate! ***Άλλο παράδειγμα:***

```
<< Graphics`FilledPlot`
FilledPlot[{0, y^2}, {y, 0, 3}, AxesOrigin → {0, 0},
 AspectRatio → 1 / 2, AxesLabel → {"y", "x"}]
```



- Graphics -

εδώ $D = \{(x, y) \in \mathbb{R}^2 / 0 \le y \le 3,\ 0 \le x \le y^2\}$ *και ολοκληρώνουμε πρώτα ως προς* x *και μετά ως προς* y *δηλ.*

```
Integrate[2 x - 3 y^2, {y, 0, 3}, {x, 0, y^2}]
```

$-\frac{486}{5}$

Το AspectRatio→1/2 σημαίνει όπως έχουμε αναφέρει και παλία ότι το μήκος του κάθετου πλαισίου που περιβάλλει το διάγραμμα είναι το μισό του μήκους του οριζοντίου. *Γενικά να έχουμε στο νού ότι ολοκληρώνουμε* ***τελευταία*** *σε διάστημα που έχει σταθερά άκρα δηλ. είναι της μορφής του κλειστού διαστήματος* $[\alpha,\beta]$. *π.χ*

Έχουμε την f[x,y,z]:=x y^2 z^3 *και θέλουμε να βρούμε το τριπλό ολοκλήρωμα στο χωρίο* $D = \{(x, y, z) \in \mathbb{R}^3 / 0 \le x \le 1, 0 \le y \le 1-x,\ 0 \le z \le x*y\}$ *που περικλείεται απο τα επίπεδα* x+y=1 *και* z=0, *και την επιφάνεια(υπερβολικό παραβολοειδές)* z=x*y . *Λόγω της μορφής του* D *ολοκληρώνουμε πρώτα ως προς* z *μετά ως προς* y *και* ***τέλος*** *ως προς* x.

```
Integrate[x y^2 z^3, {x, 0, 1}, {y, 0, 1 - x}, {z, 0, x y}]
```

$\frac{1}{288288}$

Βέβαια το ίδιο ολοκλήρωμα θα μπορούσαμε να το κάνουμε διαδοχικά σε τρία βήματα αλλά φυσικά είναι κουραστικό...

```
int1 = Integrate[x y^2 z^3, {z, 0, x y}]
int2 = Integrate[int1, {y, 0, 1 - x}]
int3 = Integrate[int2, {x, 0, 1}]
```

$$\frac{x^5 y^6}{4}$$

$$\frac{1}{28} (1-x)^7 x^5$$

$$\frac{1}{288288}$$

Παρακάτω θα προσπαθήσουμε να σχεδιάσουμε στο τρισδιάστατο χώρο ***το*** D για να πάρουμε μια γεύση απο τρισδιάστατα γραφικά. Για περισσότερα μπορείτε να πάτε στο κεφάλαιο για τις γραφικές παραστάσεις. ***Για βοήθεια πάνω στη γραφική συνάρτηση*** ContourPlot3D ***που χρησιμοποιούμε μπορείτε να ανατρέξετε στο*** Help ***και επίσης να "παίξετε" με τις γραφικές παραστάσεις αφαιρώντας*** ή ***αλλάσοντας κάποια χαρακτηριστικά***(options π.χ. τα Ticks, Axeslabel) ***ώστε να καταλάβετε τι ακριβώς κάνουν***. ***Θα πούμε μονάχα*** ότι ***η*** plot1 ***παριστάνει το επίπεδο*** z==0 ***με κόκκινο χρώμα***(***Σχόλιο***: ***αλλάξαμε το*** z==0 (Το κάτω άκρο της μεταβλητής z)***σε κάτι σχεδόν όμοιο*** z-0.00001==0 ***διότι δεν μπορέσαμε να χρωματισουμε το επίπεδο*** z==0 ***με την*** ContourPlot3D!!) ***η*** plot2 ***παριστάνει το επίπεδο*** x+y=1 ***σε πράσινο και η*** plot3 ***το παραβολοειδές*** z=x y(το πάνω άκρο για την z) ***με μπλέ χρώμα***. Το D ***είναι το κλειστό χωρίο που περικλείεται μεταξύ των τριών επιφανειών***.
Σχόλιο: Η Show μας επιτρέπει να δούμε στο ίδιο γράφημα και τις τρείς επιφάνειες plot1,plot2,plot3.

```
Remove[x, y, z]
<< Graphics`ContourPlot3D`
plot1 = ContourPlot3D[z - 0.00001, {x, 0, 1}, {y, 0, 1}, {z, 0, 1},
   ContourStyle → RGBColor[1, 0, 0], DisplayFunction → Identity];
plot2 = ContourPlot3D[x + y - 1, {x, 0, 1}, {y, 0, 1}, {z, 0, 1},
   ContourStyle → RGBColor[0, 1, 0], DisplayFunction → Identity];
plot3 = ContourPlot3D[z - (x y), {x, 0, 1},
   {y, 0, 1}, {z, 0, 1}, PlotPoints → {5, 5},
   ContourStyle → RGBColor[0, 0, 1], DisplayFunction → Identity];
Show[plot1, plot2, plot3, Axes → True, Ticks →
  {{{0, "(0,0,0)"}}, {{0, "(1,0,0)"}, {1, "(1,1,0)"}}, None},
 AxesLabel → {"x Axis", "y Axis", "z Axis"},
 AxesEdge → {{-1, -1}, {1, -1}, Automatic},
 Lighting → False, DisplayFunction → $DisplayFunction,
 ViewPoint → {2.434, -1.853, 0.866}]
```



- Graphics3D -

Αντί του παραπάνω plot1 *μπορείτε και να βάλετε ένα κόκκινο πολύγωνο ως εξής*

```
plot1 = Graphics3D[
  {RGBColor[1, 0, 0], Polygon[{{0, 0, 0}, {1, 0, 0}, {1, 1, 0}, {0, 1, 0}}]},
  DisplayFunction → Identity]
```

δοκιμάστε παραπάνω να δείτε την διαφορά! Για περισσότερες πληροφορίες για το Graphics3D πηγαίνετε στο Help.

Άσκηση: Να *υπολογιστεί το διπλό ολοκλήρωμα της* $f[x, y] = \frac{1}{E^x \mathrm{Sqrt}[y]}$ *στην περιοχή* $D = \{(x, y) \in \mathbb{R}^2 / -2 \leq y \leq 0, |x-1| \leq 4\}$ *και να γίνει γραφική παράσταση του* D.